

---

---

## Technical Notes

---

---

### Patching the ONLY Structure on Top of fig-FORTH

*Man Chor Ko and Scott Jenson*

*Ascent Inc.*

*1983 Landings Dr.*

*Mountain View, CA 94030*

The ONLY structure is by far the most powerful enhancement made to the FORTH-83 standard [1] [2]. It expands the power of the vocabulary structure to become a truly useful software development tool. Dictionary searches can be limited to search only a group of five words if you so desire. This allows for much more modular programming, vocabularies becoming akin to procedures in PASCAL. Scoping can be limited very easily as well, so local and global variables can have the same names.

The following screens allow you to take your current fig system and add the ONLY structure to it. We loaded it up on top of LMI's\* Z80 FORTH with CP/M file support and have used it extensively now for 6 months with no problem. The strongest advantage of this system is that even after loading up the ONLY code, CURRENT and CONTEXT are functionally equivalent to the fig model so all existing fig utilities can be loaded with no changes. The ultimate test of this claim came when we loaded NAUTILUS system's cross compiler on top of this ONLY system and it compiled without a hitch.

The only words in the listing which may be unfamiliar are CMOVE> which moves from high memory to low, instead of low to high and --, which is our in-house commenting word defined as:

```
: --  
  IN @  
  64 / 1+  
  64 * IN!  
; IMMEDIATE
```

which simply skips the rest of the screen line. Also, an ORDER will always print an ONLY in the search list. This prevents you from getting locked into a sealed vocabulary with no way out. Note: there is no error checking on TOSS. If you TOSS one too many vocabularies you will get a new system prompt of "aok". The system won't crash but WORDS or VLIST will give you garbage. You can easily recover with an ALSO FORTH.

After having used the ONLY structure for a few months, going back to the old FORTH-79 or fig models is like working with bear skins and stone knives. It is a truly useful addition to any FORTH system. I hope you enjoy it.

\*Laboratory Microsystems, Inc.

## References

- [1] Ragsdale, William F., "The ONLY Concept for Vocabularies," *Proceedings of the 1980 FORML Conference*, pp 105-115, Forth Interest Group, San Carlos, CA, 1980.
- [2] *Forth-83 Standard*, Forth Standards Team, 1984.

### Screen #9

```

0 ( Loading the ONLY structure on top of fig-FORTH )
1
2 FORTH DEFINITIONS
3 CR ." Loading VOC utilities."
4 10 12 THRU
5 CR ." Loading boot-strapping code for CONTEXT."
6 13 LOAD
7 CR ." Loading boot-strapping code for ONLY."
8 14 18 THRU
9
10 DECIMAL ;S
11
12
13
14
15

```

### Screen # 10

```

0 ( ONLY vocabulary utilities - (.VOC, ?ENDVOC          saj 5/21/84)
1
2 HEX
3 : (.VOC)          ( vlf -- )
4 4 -              -- get to the PFA
5 NFA ID.          -- then nfa and print name
6 ;
7 : ?ENDVOC        ( lfa -- lfa\flag )
8 DUP @ A081 =     -- is it a pseudo name field?
9 OVER 0= OR      -- or a sealed voc? then return true
10 ;
11 DECIMAL
12
13
14
15

```

## Screen # 11

```

0 ( VOC UTILITIES - WORDS                saj 5/21/84)
1 HEX
2 : WORDS          ( -- )          -- list words in CONTEXT only
3 50 OUT !
4 CONTEXT @ @          -- NFA of last word in vocab
5 BEGIN
6  ?ENDVOC          -- while not end of vocab
7  ?TERMINAL OR 0=   --   or request to quit
8 WHILE
9   OUT @ C/L 0C - >   -- first, are we at coln?
10  IF CR 0 OUT ! ENDIF -- yes. .next line
11  DUP ID. SPACE SPACE -- print out name and some space
12  PFA LFA @         -- get next word above
13 REPEAT DROP
14 ;
15 DECIMAL

```

## Screen # 12

```

0 ( ONLY -- SEAL                saj 5/21/84)
1
2 : SEAL          ( -- )
3 CONTEXT @          -- start looking here
4 BEGIN
5   DUP @ ?ENDVOC ?TERMINAL OR 0=   -- link through until end
6 WHILE DROP PFA LFA
7 REPEAT DROP 0 SWAP !          -- store 0 in LFA
8 ;
9 DECIMAL
10
11
12
13
14
15

```

## Screen # 13

```

0 ( Patching context into a stack      saj 5/21/84)
1
2 6 CONSTANT MAXVOC          -- max vocabs in search list
3 MAXVOC 1- 2* CONSTANT .LASTVOC -- offset to last vocab
4
5 : PATCH          -- turns CONTEXT into constant
6 HERE CONTEXT @ ,          -- which points to an array
7 ' CONTEXT !
8 ' C/L CFA @ ' CONTEXT CFA !
9 MAXVOC 1- 2* ALLOT
10 ;
11
12 PATCH
13
14
15

```

## Screen # 14

```

0 ( ONLY and ALSO                                saj 5/21/84)
1 VOCABULARY ONLY  ONLY DEFINITIONS
2 : ALSO                                           -- slide transient into resident
3 CONTEXT . .LASTVOC + 2-                          -- addr of last resident
4 @ -DUP IF                                         -- last resident < 0
5 (.VOC) ." removed from CONTEXT."
6 ENDIF
7 CONTEXT DUP 2+ [ MAXVOC 2- 2* ] LITERAL CMOVE>
8 ;
9
10 HERE 2+ ]                                         -- alter runtime of ONLY
11 DOES>
12   CONTEXT . .LASTVOC ERASE                       -- removes resident+trans.
13   2+ DUP CONTEXT ! CONTEXT . .LASTVOC + !
14   EXIT [
15 ' ONLY !                                         -- patch ONLY's ptr -> runtime code

```

## Screen # 15

```

0 ( TOSS                                          saj 5/21/84)
1 HEX
2 : TOSS                                           -- move residents up, remove transient
3 CONTEXT DUP 2+ SWAP [ MAXVOC 2- 2* ] LITERAL CMOVE
4 0 CONTEXT . .LASTVOC + 2- !
5 ;
6 : X                                             -- null for term. and interp. control
7 6I EMIT
8 R> DROP
9 ;
10 80C1 ' X NFA !                                 -- patch in a null header
11 DECIMAL
12
13
14
15

```

## Screen # 16

```

0 ( ONLY -- search order control words          saj 5/21/84)
1 DECIMAL
2 : FORTH [COMPILE] FORTH ; IMMEDIATE
3 : DEFINITIONS DEFINITIONS ;
4 : FORGET FORGET ;
5 : VOCABULARY VOCABULARY ;
6 : ONLY ONLY ;
7 : WORDS WORDS ;
8
9 : ORDER                                         -- show the search order in CONTEXT vector
10 CONTEXT MAXVOC 2* OVER + SWAP
11 DO 1 @ -DUP IF (.VOC) ENDIF
12   2 +LOOP
13 10 SPACES CURRENT @ (.VOC)
14 ;
15

```

## Screen # 17

```

0 ( -FIND - search CONTEXT vector          saj 5/21/84)
1 0 VARIABLE AVOC
2 : (IVOC)                                -- search only one sealed vocab
3 -DUP IF @ (FIND) -DUP
4     IF R> R> 2DROP ELSE 2 AVOC +! ENDIF
5     ELSE DROP 2 AVOC +! ENDIF
6 ;
7 : -FIND ( -- PFA\LEN\T )
8   BL WORD
9   CONTEXT AVOC !                        -- first vocab
10  BEGIN
11   HERE AVOC @ @ (IVOC)                 -- search 1 voc.
12   AVOC @ [ CONTEXT MAXVOC 2* + ] LITERAL = -- end of CONTEXT?
13  UNTIL
14  0 R> DROP                             -- unwind patch
15 ;

```

## Screen # 18

```

0 ( Patch nucleus -FIND & new :          saj 5/21/84)
1 LATEST PFA CFA
2 FORTH ' -FIND !                        -- replace system -FIND
3 ONLY
4
5 ' CURRENT CFA                          -- patch : to set CURRENT equal to
6 ' : 8 + !                              -- CURRENT
7
8
9 ONLY SEAL FORTH DEFINITIONS -- make FORTH transient, current
10 ;S

```

